
Abstract

A signi�cant portion of the ANSI/IEEE-854 Standard for Radix-Independent Floating-Point Arith-
metic is de�ned in PVS (Prototype Veri�cation System). Since IEEE-854 is a generalization of the
ANSI/IEEE-754 Standard for Binary Floating-Point Arithmetic, the de�nition of IEEE-854 in PVS
also formally de�nes much of IEEE-754. This collection of PVS theories provides a basis for ma-
chine checked veri�cation of 
oating-point systems. This formal de�nition illustrates that formal
speci�cation techniques are su�ciently advanced that it is reasonable to consider their use in the
development of future standards.

keywords: Floating-point arithmetic, Formal Methods, Speci�cation, Veri�cation.



1 Introduction

This document describes a de�nition of the ANSI/IEEE-854 [3] Standard for Radix-Independent
Floating-Point Arithmetic in the PVS veri�cation system (developed at SRI International) [4].
IEEE-854 is a generalization of the ANSI/IEEE-754 [2] Standard for Binary Floating-Point Arith-
metic. Therefore, this formalization of the IEEE-854 can be instantiated to serve as a basis for the
formal speci�cation of the more widely used IEEE-754 standard. All that is required is to instan-
tiate the general theory with the appropriate constants, and de�ne the representation formats in
accordance with IEEE-754.

This is not the �rst formalization of an IEEE standard for 
oating-point arithmetic. Geo�
Barrett [1] describes the Z formalization of IEEE-754 used in the development of the INMOS
T800 Transputer. Z is a formal speci�cation language with limited mechanized support. The
speci�cation presented here uses the PVS speci�cation language which is tightly integrated with
the PVS mechanized proof system. Also, the speci�cation presented here is of IEEE-854, not
IEEE-754. This formalization in PVS was not based upon the Z speci�cation.

This document will present those portions of the standard that have been de�ned in PVS. The
various features of PVS will be described at the time of their �rst use. This report highlights some
areas of imprecision in the standard and illustrates that formal techniques are su�ciently advanced
to consider their use in the development of future standards.

2 Basic De�nitions

The document IEEE-854 (hereafter referred to as the standard) describes a parameterized standard
for 
oating-point arithmetic. This section will present the de�nition of 
oating-point numbers and
introduce mappings between 
oating-point and real numbers. The standard allows the de�nition
of four precisions of 
oating-point numbers: single, single extended, double and double extended.
Each precision is distinguished by the range of representable values and the number of signi�cant
digits. The PVS theories de�ne a �xed, but undetermined precision. It is simple to de�ne any
combination of precisions by importing multiple instances of the top-level PVS theory presented
here.

2.1 Sets of Values

Section 3.1 of the standard de�nes the parameters:

Four integer parameters specify each precision:

b = the radix

p = the number of base-b digits in the signi�cand

Emax = the maximum exponent

Emin = the minimum exponent

The parameters are subject to the following constraints:

1. b shall be either 2 or 10 and shall be the same for all supported precisions

2. (Emax �Emin)=p shall exceed 5 and should exceed 10

3. bp�1 � 105

The balance between the over
ow threshold (bEmax+1) and the under
ow threshold (bEmin)
is characterized by their product (bEmax+Emin+1), which should be the smallest integral
power of b that is � 4. [3, page 8]

From these constraints, it is clear that b > 1, p > 1, and that Emax > Emin. However, the last
quoted sentence on balance between the over
ow and under
ow thresholds is only a suggestion1

and need not be followed for an implementation to be compliant. In later sections, we will highlight
some consequences of not having a balanced exponent range.

In PVS, these constraints can be de�ned as follows:

IEEE_854 [b,p:above(1),E_max,E_min:integer]: THEORY

BEGIN

ASSUMING

Base_values: ASSUMPTION b=2 or b=10

Exponent_range: ASSUMPTION (E_max - E_min)/p > 10

Significand_size: ASSUMPTION b^(p-1)>=10^5

E_balance: ASSUMPTION

IF b < 4 THEN E_max + E_min = 1 ELSE E_max + E_min = 0 ENDIF

ENDASSUMING

Exponent_balance: LEMMA b^(E_max+E_min) <4 & 4<=b^(E_max+E_min+1)

E_max_gt_E_min: LEMMA E_max > E_min

E_min_neg: LEMMA E_min<0

E_max_pos: LEMMA E_max>0

IMPORTING IEEE_854_defs[b,p,E_max,E_min]

END IEEE_854

This theory de�nition has four formal parameters, which correspond to the requirements of the
standard. For a �xed n, the type above(n) is de�ned in the PVS prelude as fi : natji > ng, thus
by declaring b and p to be of type above(1), we have b > 1 and p > 1. The PVS ASSUMING section
states the additional constraints on these parameters. These assumptions de�ne proof obligations
for any theory that imports IEEE 854. This speci�cation includes the optional constraints given by
the standard. A minimally compliant speci�cation would modify assumption Exponent range and
remove both assumption E balance and lemmas Exponent balance, E min neg, and E max pos.
The last line of the PVS theory imports the remaining de�nitions and declarations to complete
the speci�cation of 
oating-point arithmetic for a �xed precision (e.g. one of single, double, single
extended, or double extended).

None of the underlying de�nitions depend directly on the assumptions given in the assuming
section, so the theories de�ning the rest of the standard will use the weaker assumptions that b > 1,
p > 1, and that Emax > Emin. We will not assume that Emax � 0 or that Emin � 0, since these

1The distinction is between should and shall. Usage of the word should indicates a suggestion as opposed to a
requirement.

are not consequences of the required constraints. This will have a limited impact on the remaining
de�nitions.

Section 3.1 of the standard continues:

Each precision allows for the representation of just the following entities:

1. Numbers of the form (�1)sbE(d0:d1d2 � � �dp�1) where

s = an algebraic sign

E = any integer between Emin and Emax, inclusive

di = a base-b digit (0 � di � b� 1)

2. Two in�nities, +1 and �1

3. At least one signaling NaN

4. At least one quiet NaN [3, page 8]

Item 1 has a slight ambiguity concerning the de�nition of s. If s is de�ned as an algebraic sign
(e.g. one of f+;�g), then the expression (�1)s has no meaning. The PVS speci�cation adopts the
de�nition from IEEE-754 [2], that is, s 2 f0; 1g. This is the most natural choice for s, but several
other numeric encodings possess the necessary properties. For example, s could be a base-b digit
where an even value denotes positive and an odd value denotes negative.

The PVS speci�cation of values de�nes a 
oating-point number (fp num) using the PVS abstract
datatype mechanism [5]2:

IEEE_854_values

[b,p:above(1),

E_max:integer,

E_min:fi:integer | E_max > ig]: THEORY

BEGIN

sign_rep: type = fn:nat | n = 0 or n = 1g
Exponent: type = fi:int | E_min <= i & i <= E_maxg
digits: type = [below(p)->below(b)]

NaN_type: type = fsignal, quietg
NaN_data: NONEMPTY_TYPE

fp_num: datatype

begin

finite(sign:sign_rep,Exp:Exponent,d:digits):finite?

infinite(i_sign:sign_rep): infinite?

NaN(status:NaN_type, data:NaN_data): NaN?

end fp_num

2This theory has no assuming section, but there is an explicit assumption in the formal parameters. Emin is
de�ned via the dependent type mechanism to be strictly less than Emax. By capturing this information in the type
of Emin, the corresponding importing tccs are trivially satis�ed (and hence, not generated). If we used an assuming
clause, there would be an explicit proof obligation at each level of the importing hierarchy.

[...]

The de�nition of datatype fp num states that the type of 
oating-point numbers is the disjoint
union of three sets: �nite numbers, in�nite numbers, and Not a Numbers (NaNs). A �nite number
can be constructed (using constructor finite) from an algebraic sign, an integer exponent, and a
signi�cand3 ; an in�nity can be constructed from an algebraic sign; and a NaN can be constructed
from a status 
ag (i.e. signal or quiet) and data undetermined by the standard.

2.2 Mapping 
oating-point numbers to reals

The standard implies an intended semantics for the representable numeric values. The function
value maps the �nite 
oating point numbers to the reals as implicitly speci�ed in the standard. In
PVS, reals are treated as a base type. There is no need to import a library of de�nitions for real
arithmetic.

fin : var (finite?)

value_digit(d:digits)(n:nat):nonneg_real =

if n < p then d(n) * b ^ (-n) else 0 endif

value(fin): real =

(-1) ^ sign(fin) * b ^ Exp(fin) * Sum(p, value_digit(d(fin)))

Here, fin is declared to be a variable of type (finite?), that is, an element of the subtype of
fp num that satis�es the predicate finite?4. Function value digit takes a collection of digits5

and an index for a particular digit and returns the usual base-b interpretation of a digit determined
by its position in the signi�cand. Finally, value de�nes the interpretation of the sign �eld, the
exponent, and sums the values of the digits in the signi�cand. Function Sum is de�ned in a separate
PVS theory.

The standard recognizes that this scheme encodes some values redundantly. Furthermore,
an implementation may use redundant encodings, so long as it does not distinguish redundant
encodings of nonzero numbers. The standard subdivides the encodings into three groups using the
following de�nitions:

normal number A nonzero number that is �nite and not subnormal.

subnormal number A nonzero 
oating-point number whose exponent is the preci-
sion's minimum and whose leading signi�cant digit is zero. [3, Section 2, page
8]

These de�nitions divide the �nite numbers into three groups: those that denote zero (i.e. any �nite
fp num with a signi�cand of all zeros), the subnormal numbers, and the normal numbers. The
de�nitions given above are imprecise. Consider the following �nite number:

(�1)0 � bEmin+1 � 0:01 � � �0 (= bEmin�1)

3The signi�cand consists of an indexed collection of p base-b digits. The most natural way to de�ne this in PVS
is by function type [below(p)! below(b)]. The PVS prelude de�nes below(n) : type = fi : natji < ng.

4The predicate finite? is determined by the fp num datatype declaration.
5The PVS speci�cation language is an extended version of higher-order logic. Functions are �rst-class objects and

can be passed as parameters or be de�ned as the return type of other functions.

This is clearly nonzero. Since the exponent is not the precision's minimum, a strict interpretation of
the de�nition of subnormal may lead us to conclude that this must be a normal number. However,

bEmin�1 = (�1)0 � bEmin � 0:1 � � �00

thus it is equal to a subnormal number so it must also be subnormal. Before we can make the
distinctions precise, we need to determine a canonical encoding for each �nite 
oating-point number.
The function normalize maps each �nite fp num to a canonical representative:

shift_left(d: digits): digits =

LAMBDA (i: below(p)): IF (i + 1 = p) THEN 0 ELSE d(i + 1) ENDIF

normalize(fin:(finite?)): recursive (finite?) =

IF Exp(fin) = E_min or d(fin)(0) /= 0 then

fin

ELSE

normalize(finite(sign(fin),Exp(fin)-1,shift_left(d(fin))))

ENDIF

measure lambda (fin:(finite?)): Exp(fin) - E_min

Recursive6 function normalize repeatedly shifts the signi�cand left one digit and decrements the
exponent until either the exponent is the precision's minimum or the most signi�cant digit is
nonzero. Since our goal in de�ning function normalize is to map each �nite number to a canonical
representative, we must prove that the result of this function preserves the value of its argument.
The following lemma has been proven in PVS:

normal_value: LEMMA

value(fin) = value(normalize(fin))

We can now make the distinctions between �nite numbers precise. We do this by de�ning three
predicates: zero?, normal?, and subnormal?.

zero?(fp:fp_num):bool =

IF finite?(fp) THEN value(fp)=0 ELSE FALSE ENDIF

normal?(fp: fp_num): bool =

IF finite?(fp) THEN d(normalize(fp))(0) > 0 ELSE FALSE ENDIF

subnormal?(fp: fp_num): bool =

IF finite?(fp) THEN not zero?(fp) &

Exp(normalize(fp)) = E_min &

d(normalize(fp))(0) = 0

ELSE FALSE

ENDIF

We can provably partition the �nite numbers into three sets:

6PVS requires that all functions be total, so any de�nition by recursion involves a proof that the recursion
terminates. The evidence needed for this proof is given by a measure function that must decrease in each recursive
call according to a well-founded relation. The default well-founded relation is `<' de�ned on the natural numbers.

finite_cover: LEMMA zero?(fin) OR normal?(fin) OR subnormal?(fin)

finite_disjoint1: LEMMA NOT (zero?(fin) & normal?(fin))

finite_disjoint2: LEMMA NOT (zero?(fin) & subnormal?(fin))

finite_disjoint3: LEMMA NOT (normal?(fin) & subnormal?(fin))

Lemma finite cover states that every �nite 
oating-point number either zero, normal, or sub-
normal. The remaining lemmas assert that these sets are mutually disjoint.

There are several simple lemmas that can be proven about value. We introduce the following
de�nitions for the maximum and minimum representable values within a precision:

max_significand:digits =

(lambda (i:below(p)): b-1)

min_significand: digits =

(lambda (i: below(p)): IF i < p - 1 THEN 0 ELSE 1 ENDIF)

d_zero: digits = lambda (i: below(p)): 0

pos : sign_rep = 0

neg : sign_rep = 1

max_fp_pos : fp_num = finite(pos,E_max,max_significand)

min_fp_pos : fp_num = finite(pos,E_min,min_significand)

pos_zero : fp_num = finite(pos,E_min,d_zero)

With these de�nitions we can prove in PVS that the function value returns the correct value for
these 
oating-point numbers.

max_fp_correct: LEMMA

value(max_fp_pos) = b ^ (E_max + 1) - b ^ (E_max - (p - 1))

min_fp_correct: LEMMA

value(min_fp_pos) = b ^ (E_min - (p - 1))

value_of_zero: LEMMA

value(pos_zero) = 0

Function value only speci�es part of the relationship between reals and 
oating-point numbers.
It serves to interpret �nite 
oating-point numbers as reals. The next section addresses mapping
reals to 
oating-point numbers.

2.3 Mapping reals to 
oating-point numbers

To map reals to 
oating-point, we de�ne the following functions:

sign_of(r:real): sign_rep = IF r < 0 THEN neg ELSE pos ENDIF

Exp_of(px:posreal): fi:int| b^i <= px & px < b^(i+1)g
truncate(E:integer,nnx:nonneg_real): digits =

(lambda (i:below(p)): mod(floor(nnx/(b^(E-i))),b)

Function sign of returns the algebraic sign of a real number, adopting the convention that the
sign of 0 is positive. Function Exp of is completely de�ned using the dependent type and predicate
subtype features of PVS. The range type of Exp of depends on its argument px, and is constrained
to be an integer that satis�es the given predicate. PVS generates a type-correctness condition
(TCC) for this de�nition.7 The TCC is discharged by showing that for all positive reals px there
is an integer that satis�es the predicate. Function truncate uses the mod and floor functions to
determine each digit in the signi�cand for a given exponent E. These three functions allow us to
de�ne the following conversion from real numbers to 
oating-point numbers:

real_to_fp(r): fp_num =

IF abs(r) >= b^(E_max+1) THEN

infinite(sign_of(r))

ELSIF abs(r) < b^E_min THEN

finite(sign_of(r), E_min, truncate(E_min, abs(r)))

ELSE

finite(sign_of(r), Exp_of(abs(r)), truncate(Exp_of(abs(r)), abs(r)))

ENDIF

Function real to fp converts an arbitrary real into a 
oating-point representation. If the real is
outside the range of representable values, an appropriately signed in�nity is returned. If the real
is too small to be represented, it gets mapped to an appropriately signed zero. This de�nition
provides an approximation of reals by rounding toward zero. However, the standard calls for four
di�erent rounding modes. The next section describes the various rounding mode and shows how
this de�nition may be used in a general conversion from reals to 
oating-point.

3 Rounding

Floating-point numbers serve as a computable approximation of real numbers. The standard speci-
�es four means of approximating reals by 
oating-point numbers. The user has the ability to select
the rounding mode from among these four. In Section 4, the standard states:

. . .every operation speci�ed in section 5 shall be performed as if it �rst produced an
intermediate result correct to in�nite precision and with unbounded range, and then
that result rounded according to one of the modes in this section. [3, page 9]

The operations in section 5 referenced by this clause consist of the basic arithmetic operations:
add, subtract, multiply, divide; remainder and square root; comparisons; conversions between
precisions; and conversions to integers and integer valued 
oats. The discussion of rounding modes
will be done in conjunction with the speci�cation of the operations given in section 5 of the standard.
The default mode is round to nearest. The standard states:

An implementation of this standard shall provide round to nearest as the default round-
ing mode. In this mode the representable value nearest to the in�nitely precise result
shall be delivered; if the two nearest representable values are equally near, the one with
its least signi�cant digit even shall be delivered. [3, Section 4.1, page 9]

In addition, the standard continues:

7A TCC is a proof obligation generated by PVS that is su�cient to show that a given term is well typed. PVS
has an undecidable type system, so sometimes the user must provide a proof that a term is correctly typed.

An implementation of this standard shall also provide three user-selectable directed
rounding modes: round towards +1, round towards �1, and round towards 0. [3,
Section 4.2, page 9]

These rounding modes will �rst be de�ned for conversion of reals to integers.

3.1 Floating-point to integer

The simplest rounding operation to consider is converting a 
oating-point number to an integer (or
to an integral valued 
oating-point number). From Section 2.1, we de�ned function value to map
a 
oating-point number to a real. All that remains is to convert this real to an integer. Section 5.4
of the standard states:

Conversion to integer shall be e�ected by rounding as speci�ed in Section 4. [3, page
10]

Section 5.5 adds:

It shall be possible to round a 
oating-point number to an integral valued 
oating-point
number in the same precision. [3, page 10]

The four rounding modes are speci�ed as an enumerated type in PVS, leading to the following
de�nition of function round:

sgn(r:real): int = IF r >= 0 THEN 1 ELSE -1 ENDIF

round(r:real,mode:rounding_mode): integer =

CASES mode of

to_nearest: round_to_even(r),

to_zero: sgn(r) * floor(abs(r)),

to_pos: ceiling(r),

to_neg: floor(r)

ENDCASES

This de�nition makes use of the 
oor, ceiling, and absolute value functions to de�ne the directed
roundings. Round to nearest requires an additional function de�nition.

round_to_even(r:real): integer =

IF r - floor(r) < ceiling(r) - r THEN floor(r)

ELSIF ceiling(r) - r < r - floor(r) THEN ceiling(r)

ELSIF floor(r) = ceiling(r) THEN floor(r)

ELSE 2 * floor(ceiling(r) / 2)

ENDIF

Function round to even rounds an arbitrary real to the nearest integer. The typical cases are
de�ned using the integer 
oor and ceiling functions. The di�cult case is the fourth alternative
where the fractional part of r is 1/2. The expression

2�

�
dre

2

�

will round any real number r to the nearest even integer.
To demonstrate the correctness of these de�nitions, the following lemmas have been proven in

PVS:

round_to_even1: LEMMA

abs(r - round_to_even(r)) <= 1 / 2

round_to_even2: LEMMA

abs(r - round_to_even(r)) = 1 / 2

=> integer_pred(round_to_even(r) / 2)

round1: LEMMA abs(r - round(r,mode)) < 1

Two of these lemmas illustrate the correctness of the de�nition of round to even. The �rst states
that round to even(r) has an approximation error of at most 1=2. The second states that when
it is in error by exactly 1=2, round to even returns an even integer. In addition, Lemma round1

illustrates the correctness of round. The approximation error is always less than 1.
We can use round to de�ne a function mapping a �nite 
oating-point to an integer (and to an

integral valued 
oating-point number):

fp_to_int(fin,mode):integer = round(value(fin),mode)

fp_to_int_fp(fin,mode): fp_num =

real_to_fp(round(value(fin),mode))

Ideally, function fp to int fp should return an object of type (finite?). However, since there
are no constraints to ensure that Emax � p, it is not possible to prove the resulting TCC.8

3.2 Rounding reals

The function round can be used in conjunction with real to fp to de�ne a general rounding
function in accordance with the standard. It is necessary to �rst scale a nonzero real so that
its scaled value is between bp�1 and bp. Function round can be used to adjust the scaled value
accordingly, and the result can be scaled back to its original magnitude. The resulting real will
have at most p signi�cant base-b digits, so real to fp can be used to map it into an appropriate

oating-point representation. The standard describes a number of di�erent possible return values
if r is outside the range of normal 
oating-point numbers. It is not practical to introduce all the
possible scenarios here. The exceptional cases will be presented in a later section of this paper.

scale(px):fi:int|b^(i+p-1)<=px & px < b^(i+p)g = Exp_of(px)-(p-1)

scale_correct: lemma b^(p-1)<=px/b^scale(px) & px/b^scale(px)< b^p

over_under?(r): bool = (r/=0 & (abs(r)>max_pos or abs(r)<b^E_min))

round_scaled(r:nzreal,mode:rounding_mode): real =

b^(scale(abs(r)))*round( b^(-scale(abs(r)))*r,mode)

fp_round(r, mode): real =

8If Emax < 0, then the maximum representable 
oating-point number may round to 1. Function real to fp maps
1 to +1 in this case. More generally, if Emax < p, a 
oating-point converted to an integer using the rounding modes
may map back to 1. From this observation, one may conclude that a reasonable instance of this standard should
adhere to the suggested constraint on a balanced exponent range.

IF r = 0 THEN 0

ELSIF over_under?(r) then

round_exceptions(r,mode)

ELSE round_scaled(r,mode)

ENDIF

Function scale has a type de�ned in the same manner as Exp of (Section 2.3), however, it also
has a body explicitly de�ning the function. Function fp round performs the necessary scaling to
appropriately round an arbitrary real. The following lemmas show that these de�nitions have the
desired properties:

round_0: LEMMA fp_round(0, mode) = 0

round_error: LEMMA

r /= 0 & NOT over_under?(r)

=> abs(r - fp_round(r, mode))

< b ^ (Exp_of(abs(r)) - (p - 1))

round_near: LEMMA

r /= 0 & NOT over_under?(r)

=> abs(r - fp_round(r, to_nearest))

<= b ^ (Exp_of(abs(r)) - (p - 1)) / 2

round_pos: LEMMA

NOT over_under?(r) => fp_round(r, to_pos) >= r

round_neg: LEMMA

NOT over_under?(r) => fp_round(r, to_neg) <= r

round_zero: LEMMA

NOT over_under?(r)

=> abs(fp_round(r, to_zero)) <= abs(r)

Lemma round 0 shows that rounding 0 returns 0 regardless of the rounding mode. Lemma
round error shows that the approximation error is less than one \least signi�cant digit". Lemma
round near states that the approximation error for mode to nearest is � one-half the least sig-
ni�cant digit. Lemmas round pos, round neg, and round zero show that fp round rounds in the
proper direction for these modes.

4 Operations

Section 5 of the standard states:

All conforming implementations of this standard shall provide operations to add, sub-
tract, multiply, divide, extract the square root, �nd the remainder, . . .

. . . , each of the operations shall be performed as if it �rst produced an intermediate result
correct to in�nite precision and with unbounded range, and then coerced this intermediate
result to �t in the destination's precision. [3, page 10]

Based on this description, the PVS description will de�ne the operations using the corresponding
real arithmetic functions.

4.1 Arithmetic de�nitions

The PVS de�nitions of the four basic arithmetic operations are similar. The enumerated type

fp_ops : type = fadd, sub, mult, divg

simpli�es the de�nition of these functions. The basic de�nition for an arithmetic operation is
illustrated by the following de�nition for fp add; the de�nitions for fp sub and fp mult are nearly
identical.

fp_add(fp1, fp2, mode): fp_num =

IF finite?(fp1) & finite?(fp2) THEN fp_op(add, fp1, fp2, mode)

ELSIF NaN?(fp1) OR NaN?(fp2) THEN fp_nan(add, fp1, fp2)

ELSE fp_add_inf(fp1, fp2)

ENDIF

The function de�nition invokes one of three functions depending on the arguments. If both argu-
ments are �nite, then this function invokes the corresponding real function applied to the values
of the arguments. If one argument is a NaN, then the rules for operations on NaNs are invoked.
When one of the arguments is in�nite, the result required by the standard is returned. Each of
these cases will be described in more detail in the following sections.

The de�nition of division is a little more complicated, in that division by zero requires special
treatment:

fp_div(fp1, fp2, mode): fp_num =

IF finite?(fp1) & finite?(fp2)

THEN IF zero?(fp2)

THEN IF zero?(fp1)

THEN invalid %raise invalid

ELSE infinite(mult_sign(fp1, fp2)) %raise divide_by_zero

ENDIF

ELSE fp_op(div, fp1, fp2, mode)

ENDIF

ELSIF NaN?(fp1) OR NaN?(fp2) THEN fp_nan(div, fp1, fp2)

ELSE fp_div_inf(fp1, fp2)

ENDIF

If the second argument is zero and the �rst is not, then the function returns an appropriately signed
in�nity (and will later be modi�ed to raise the divide by zero exception). If both operands are zero,
the invalid exception is raised (here denoted by an arbitrary NaN named invalid). Otherwise,
division reduces to the same basic format as the other operators.

4.1.1 Arithmetic with �nite operands

When both operands are �nite, the formal speci�cation of 
oating-point arithmetic consists of
converting the �nite 
oating-point numbers to real numbers, performing the appropriate arithmetic
function, and then converting the resulting real number back to 
oating-point format. The following
PVS text accomplishes this:

apply(op,fin1,(fin2:fin| div?(op) => not zero?(fin))): real =

cases op of

add: value(fin1) + value(fin2),

sub: value(fin1) - value(fin2),

mult: value(fin1) * value(fin2),

div: value(fin1) / value(fin2)

endcases

fp_op(op, fin1, (fin2: fin | div?(op) => NOT zero?(fin)), mode): fp_num =

LET r = fp_round(apply(op, fin1, fin2), mode)

IN IF r = 0 THEN signed_zero(op, fin1, fin2, mode)

ELSE real_to_fp(r)

ENDIF

Function fp op does the appropriate conversions and calls function apply to perform the appro-
priate arithmetic operation. If the rounded result is zero, function signed zero (Section 4.1.4) is
invoked to return a correctly signed zero. Otherwise, real to fp converts the result to a 
oating-
point number. Function apply uses the dependent type and predicate subtype mechanisms of
PVS to restrict the domain of its third argument to nonzero numbers when the operation is div.
Without this type restriction, it would not be possible to justify the use of `/' in the de�nition of
apply.

4.1.2 Arithmetic on In�nities

The standard de�nes well behaved operations involving in�nite arguments. It states:

In�nity arithmetic shall be construed as the limiting case of real arithmetic with operands
of arbitrarily large magnitude, when such a limit exists. In�nities shall be interpreted
in the a�ne sense, that is, �1 < (every �nite number) < +1. [3, Section 6.1]

This requires special treatment for each arithmetic operator, the example given here is for 
oating-
point addition.

fp_add_inf(num1, (num2: num | infinite?(num1) OR infinite?(num))): fp_num

=

IF infinite?(num1) & infinite?(num2) THEN

IF (i_sign(num1) = i_sign(num2)) THEN num1

ELSE invalid

ENDIF

ELSIF infinite?(num1) THEN num1

ELSE num2

ENDIF

Function fp add inf takes two numeric arguments (i.e. either �nite or in�nite, but not NaN),
one of which must be an in�nity. If only one argument is an in�nity, that argument is the return
value. If both are in�nite and have the same sign, then either argument is the correct return value.
However, if the two in�nite arguments have di�erent signs, the invalid exception must be signaled.
In the de�nition here, fp add inf returns a NaN value invalid. This will serve as a placeholder
until the PVS speci�cation is revised to properly deal with exceptions. In�nity arithmetic for the
other operators is de�ned similarly.

4.1.3 Arithmetic on NaNs

The standard does not specify interpretation of NaNs, however, it does constrain the behavior
of operations given NaN arguments. The main motivation for these constraints is to enable use
of NaNs either for diagnostic information, or for implementation dependent enhancements to the
operations.

Section 6.2 of the standard states:

Every operation involving a signaling NaN or invalid operation (7.1) shall, if no trap
occurs and if a 
oating-point result is to be delivered, deliver a quiet NaN as its result.

Every operation involving one or two input NaNs, none of them signaling, shall signal
no exception, but, if a 
oating-point result is to be delivered, shall deliver as its result a
quiet NaN, which should be one of the input NaNs. [3]

The following PVS speci�cation captures the various cases for dealing with NaN arguments. Func-
tion fp quiet is constrained via the PVS dependent type mechanism to return one of its arguments.
Function fp signal tests to see if the invalid trap is enabled; if not, a quiet NaN is returned.

fp_quiet(op,fp1,(fp2| NaN?(fp1) OR NaN?(fp2))): fnan| nan=fp1 or nan=fp2g

fp_signal(op, fp1,(fp2| NaN?(fp1) OR NaN?(fp2))): fp_num =

IF trap_enabled?(invalid_operation) THEN invalid

ELSE fp_quiet(op, mk_quiet(fp1), mk_quiet(fp2))

ENDIF

fp_nan(op, fp1, (fp2| NaN?(fp1) OR NaN?(fp2))): fp_num =

IF signal?(fp1) OR signal?(fp2) THEN fp_signal(op, fp1, fp2)

ELSE fp_quiet(op, fp1, fp2)

ENDIF

4.1.4 The Algebraic Sign

The standard speci�es a set of rules for the algebraic sign of an arithmetic result. There are
two scenarios where special care is required to get the algebraic sign correct: arithmetic involving
in�nities (including division by zero), and arithmetic operations that deliver a result of zero. The
cases involving in�nities have been addressed above. When an arithmetic function evaluates to
zero, we need to determine whether to return +0 or �0. For multiplication and division, the sign is
\+" if and only if both arguments have the same sign. For addition and subtraction, the standard
states:

When the sum of two operands with opposite signs (or the di�erence of two operands
with like signs) is exactly zero, the sign of that sum (or di�erence) shall be \+" in
all rounding modes except round toward �1, in which mode that sign shall be \�."
However, x+ x = x� (�x) retains the same sign as x even when x is zero. [3, page 13]

The above de�nition of fp op invokes signed zero for all zero results. This function does the case
analysis required to return a correctly signed 
oating-point zero.

signed_zero(op, fin1, fin2, mode): ffin | zero?(fin)g =

CASES op OF

add:

IF zero?(fin1)

& zero?(fin2) & sign(fin1) = sign(fin2) THEN fin1

ELSIF to_neg?(mode) THEN neg_zero

ELSE pos_zero

ENDIF,

sub:

IF zero?(fin1)

& zero?(fin2) & sign(fin1) /= sign(fin2) THEN fin1

ELSIF to_neg?(mode) THEN neg_zero

ELSE pos_zero

ENDIF,

mult:

IF sign(fin1) = sign(fin2) THEN pos_zero

ELSE neg_zero

ENDIF,

div:

IF sign(fin1) = sign(fin2) THEN pos_zero

ELSE neg_zero

ENDIF

ENDCASES

4.2 Remainder

Section 5.1 of the standard de�nes the remainder function as follows:

When y 6= 0, the remainder r = x REM y is de�ned regardless of the rounding mode by
the mathematical relation r = x � y � n, where n is the integer nearest the exact value
of x=y; whenever jn � x=yj = 1=2, then n is even. . . . If r = 0, its sign shall be that of
x. [3]

The function round to even, de�ned in section 3.1, gives us the necessary means to compute n

from the above description. The de�nition of the 
oating-point remainder function, fp rem, is
straightforward in PVS.

REM(fin1, (fin2:fin|not zero?(fin))): fp_num =

let x = value(fin1),

y = value(fin2) in

if (x - y * round_to_even(x/y)) = 0

then finite(sign(fin1),E_min,d_zero)

else real_to_fp(x - y * round_to_even(x/y))

endif

fp_rem(fp1, fp2): fp_num =

IF finite?(fp1) & finite?(fp2)

THEN IF zero?(fp2)

THEN invalid

ELSIF zero?(REM(fp1, fp2)) THEN finite(sign(fp1),E_min,d_zero)

ELSE REM(fp1, fp2)

ENDIF

ELSIF NaN?(fp1) OR NaN?(fp2) THEN fp_nan_rem( fp1, fp2)

ELSIF infinite?(fp1) THEN invalid

ELSE fp1

ENDIF

According to the standard, the remainder function is always exact (i.e. no rounding error). This
fact has not yet been proven in PVS.

4.3 Square Root

PVS does not have a built-in de�nition for the square root function. It can be de�ned by the
expression:

sqrt(px): fpy | py * py = pxg

This de�nition generates a TCC to prove that the range is nonempty for all positive reals px. It
carries in its type signature the relevant information about the square root function.

The speci�cation of the 
oating-point square root operation is:

fp_sqrt(fp, mode): fp_num =

IF NaN?(fp) THEN NaN_sqrt(fp)

ELSIF zero?(fp) THEN fp

ELSIF finite?(fp)

THEN IF sign(fp) = pos

THEN real_to_fp(fp_round(sqrt(value(fp)), mode))

ELSE invalid

ENDIF

ELSIF i_sign(fp) = pos THEN fp

ELSE invalid

ENDIF

4.4 Conversion between precisions

The standard does not require any combination of precisions. However, if more than one precision
is supported, the standard requires conversions between all supported precisions. A speci�ca-
tion involving multiple precisions can easily be de�ned by importing multiple instances of theory
IEEE 854. The basic operations for de�ning conversions between precisions are included in the
PVS speci�cation.

4.5 Floating-point $ decimal string

The PVS speci�cation does not yet de�ne conversions between 
oating-point numbers and deci-
mal strings. The standard places no restrictions on the decimal string format, so this cannot be
addressed fully until an implementation is de�ned.

4.6 Comparisons

The standard de�nes the comparison operations as follows:

Four mutually exclusive relations are possible: \less than," \equal," \greater than," and
\unordered." The last case arises only when at least one operand is a NaN. . . .

The result of a comparison shall be delivered in one of two ways at the implementor's
option: either as a condition code identifying one of the four relations listed above, or as
a true/false response to a predicate that names the speci�c predicate desired. [3, Section
5.7, page 12]

The �rst option is simple to de�ne in PVS. We simply extend the valuation function to provide
a value for the in�nities, and then de�ne the comparison function using the corresponding real
relations.

comparison_code: type = fgt, lt, eq, ung

fp_compare((fp1, fp2: fp_num)): comparison_code =

IF NaN?(fp1) OR NaN?(fp2) THEN un

ELSIF n_value(fp1) > n_value(fp2) THEN gt

ELSIF n_value(fp1) < n_value(fp2) THEN lt

ELSE eq

ENDIF

For each element of an enumerated type, PVS automatically generates a predicate recognizer. Thus,
we can also use the above de�nition to support our formal speci�cation of the second alternative
for realizing 
oating-point comparisons. The following are the predicate forms that the standard
requires.

% shall include

eq?(fp1,fp2) :bool = eq?(fp_compare(fp1,fp2))

ne?(fp1,fp2) :bool = not eq?(fp_compare(fp1,fp2))

gt?(fp1,fp2) :bool = gt?(fp_compare(fp1,fp2))

ge?(fp1,fp2) :bool = gt?(fp_compare(fp1,fp2)) or eq?(fp_compare(fp1,fp2))

lt?(fp1,fp2) :bool = lt?(fp_compare(fp1,fp2))

le?(fp1,fp2) :bool = lt?(fp_compare(fp1,fp2)) or eq?(fp_compare(fp1,fp2))

% should include

un?(fp1,fp2) :bool = un?(fp_compare(fp1,fp2))

All that remains is to correctly merge exception handling with the above de�nitions.

5 Exceptions

Section 7 of the standard states:

There are �ve types of exceptions that shall be signalled when detected. The signal entails
setting a status 
ag, taking a trap, or possibly doing both. With each exception should
be associated a trap under user control, as speci�ed in Section 8. . . . In some cases the
result is di�erent if a trap is enabled.

For each type of exception, the implementation shall provide a status 
ag that shall be
set on any occurrance of the corresponding exception when no corresponding trap occurs.
. . .

The only exceptions that can coincide are inexact with over
ow and inexact with un-
der
ow. [3, page 13]

The combination of exceptions and traps suggests that we need to modify the style of the PVS
speci�cation from a purely functional speci�cation to a process based speci�cation. The standard
requires that we signal exceptions only if the trap for that exception is taken. However, in the
current functional style we cannot model transfer of control to a trap handler.

This can be overcome by having each of the previously de�ned functions return a pair of values,
the second element of the pair is either an indication of the exception to be signaled, or an identi�er
to determine which trap handler to invoke.

The potential values for this identi�er are determined by the following datatype declaration:

exception: DATATYPE

BEGIN

invalid_operation : invalid?

division_by_zero : div_by_zero?

overflow : overflow?

underflow(exact: bool) : underflow?

inexact : inexact?

no_exceptions : no_exceptions?

END exception

trap_enabled?(e:exception):bool % = ?

To incorporate this strategy, the types of the previously de�ned functions need to be modi�ed
slightly. We will illustrate the changes by working through a single example, fp add. Each operation
shall now return a pair. The �rst element will be an fp num and the second will be the exception
status.

fp_add_x(fp1, fp2, mode): [fp_num, exception] =

IF finite?(fp1) & finite?(fp2) THEN fp_op_x(add, fp1, fp2, mode)

ELSIF NaN?(fp1) OR NaN?(fp2) THEN fp_nan_x(add, fp1, fp2)

ELSE fp_add_inf_x(fp1, fp2)

ENDIF

The modi�cations required for in�nity arithmetic are trivial. Section 6.1 of the standard states:

Arithmetic on 1 is always exact and therefore shall signal no exceptions, except for the
invalid operations speci�ed for 1 in Section 7.1. [3, page 13]

The de�nition for fp add inf x is:

fp_add_inf_x(num1,

(num2: num | infinite?(num1)

OR infinite?(num)))

: [fp_num, exception]

=

IF infinite?(num1) & infinite?(num2) THEN

IF (i_sign(num1) = i_sign(num2)) THEN

(num1, no_exceptions)

ELSE (invalid, invalid_operation)

ENDIF

ELSIF infinite?(num1) THEN

(num1, no_exceptions)

ELSE (num2, no_exceptions)

ENDIF

The operations on NaNs require similar modi�cations. The di�cult cases in handling exceptions
occur with the rounding operations involved in fp op x.

fp_op_x(op,fin1,(fin2:fin| div?(op) => not zero?(fin)),mode):

[fp_num, exception]

=

LET rp = fp_round_x(apply(op, fin1, fin2), mode)

IN IF proj_1(rp) = 0 THEN (signed_zero(op, fin1, fin2, mode),proj_2(rp))

ELSE real_to_fp_x(rp)

ENDIF

real_to_fp_x(r,e) : [fp_num, exception] = (real_to_fp(r),e)

Function fp round x is de�ned by:

fp_round_x(r, mode): [real,exception] =

IF r = 0 THEN (0,no_exceptions)

ELSIF over_under?(r) then

round_exceptions_x(r,mode)

ELSE (round_scaled(r,mode),is_exact?(r,mode))

ENDIF

is_exact?(r:nzreal,mode):exception =

IF round_scaled(r,mode) = r then no_exceptions ELSE inexact ENDIF

All that remains is to de�ne round exceptions x. This is a rather complicated de�nition that
breaks down into two cases. The �rst case is a potential over
ow; the second is a potential under
ow.

x: var (over_under?)

round_exceptions_x(x,mode): [fp_num, exception] =

IF abs(r)>max_pos THEN

overflow(x,mode)

ELSE underflow(x,mode)

ENDIF

Full descriptions of functions overflow and underflow will appear in the corresponding section
below.

5.1 Invalid Operation

The invalid operation exception does not require that a special value be delivered to a trap handler.
The cases where the invalid operation exception is raised for arithmetic operations have been
handled above.

5.2 Division by Zero

The standard does not require any special treatment when the trap is enabled, but it requires that
an appropriately signed in�nity be delivered when the exception is raised. The modi�ed fp div is:

fp_div_x(fp1, fp2, mode): [fp_num,exception] =

IF finite?(fp1) & finite?(fp2)

THEN IF zero?(fp2)

THEN IF zero?(fp1)

THEN (invalid, invalid_operation)

ELSE (infinite(mult_sign(fp1, fp2)), division_by_zero)

ENDIF

ELSE fp_op_x(div, fp1, fp2, mode)

ENDIF

ELSIF NaN?(fp1) OR NaN?(fp2) THEN fp_nan_x(div, fp1, fp2)

ELSE fp_div_inf_x(fp1, fp2)

ENDIF

5.3 Over
ow

The result of an over
ow is determined by both the rounding mode and over
ow trap status.

The over
ow exception shall be signaled whenever the destination precision's largest
�nite number is exceeded in magnitude by what would have been the rounded 
oating-
point result were the exponent range unbounded.

. . .

Trapped over
ows on all operations except conversions shall deliver to the trap handler
the result obtained by dividing the in�nitely precise result by b� and then rounding. [3,
page 14, section 7.3]

The standard continues by relating the possible values of � to the exponent range. The given
relation relies on the assumption that the exponent range is balanced around zero.

The over
ow threshold is di�erent for each of the rounding modes. The rounding mode also
determines the result when an over
ow occurs. The PVS de�nition is

trap_over((r1:nzreal), (r2: real), (mode: rounding_mode)): real =

IF trap_enabled?(overflow) THEN round_scaled(r1 * b ^ (-alpha), mode)

ELSE r2

ENDIF

overflow((r: nzreal | abs(r) > max_pos), (mode: rounding_mode)):

[real, exception] =

CASES mode OF

to_nearest:

IF abs(r)

>= b ^ (E_max + 1)

- (1 / 2) * b ^ (E_max + 1 - p)

THEN (trap_over(r, (sgn(r) * infinity), mode),overflow)

ELSE (sgn(r) * max_pos, inexact)

ENDIF,

to_zero:

IF abs(r) >= b ^ (E_max + 1)

THEN (trap_over(r, (sgn(r) * max_pos), mode),overflow)

ELSE (sgn(r) * max_pos, inexact)

ENDIF,

to_pos:

IF r > max_pos THEN (trap_over(r, infinity, mode),overflow)

ELSIF r <= -b ^ (E_max + 1)

THEN (trap_over(r, max_neg, mode),overflow)

ELSE (max_neg, inexact)

ENDIF,

to_neg:

IF r < max_neg THEN (trap_over(r, -infinity, mode),overflow)

ELSIF r >= b ^ (E_max + 1)

THEN (trap_over(r, max_pos, mode),overflow)

ELSE (max_pos, inexact)

ENDIF

ENDCASES

5.4 Under
ow

For results less than bEmin , we may not be able to preserve p signi�cant digits. The PVS speci�cation
includes a special rounding function for these cases of potential under
ow.

round_under((r: nzreal | abs(r) < b ^ E_min), (mode: rounding_mode)): real

= b^(E_min - (p - 1))*round(b^(-(E_min - (p - 1)))*r,mode)

The following correctness results have been proven in PVS about round under:

round_under_error: LEMMA

abs(r) < b ^ E_min

=> abs(r - round_under(r, mode)) < b ^ (E_min - (p - 1))

round_under_near: LEMMA

abs(r) < b ^ E_min

=> abs(r - round_under(r, to_nearest))

<= b ^ (E_min - (p - 1)) / 2

In addition, round under rounds in the correct direction for the directed rounding modes. This
function is the core of the de�nition of function underflow.

In the absence of traps, under
ow is signaled when a result is both tiny and inaccurate. Each
of these conditions may be de�ned in two distinct ways. Tininess occurs when a result is less than
bEmin ; it may be detected either before or after rounding. The PVS speci�cation uses the following
predicate to signal tininess:

tiny?((r: nzreal | abs(r) < b ^ E_min), (mode: rounding_mode)): bool =

IF tiny_flag THEN abs(round_scaled(r, mode)) < b ^ E_min ELSE TRUE ENDIF

Boolean constant tiny flag is used to signify which method a particular implementation is using to
signal tininess; if tiny flag = true, then tininess is detected after rounding, if tiny flag = false

we have already satis�ed the before rounding test for tininess (via the constraints on argument r).
Similarly, loss of accuracy may also be detected in one of two ways, either a loss due to denor-

malization or an inexact result. The PVS predicate detecting loss of accuracy is given by:

inaccurate?((r: nzreal | abs(r) < b ^ E_min), (mode: rounding_mode)): bool =

IF inaccurate_flag THEN (round_scaled(r,mode)/=round_under(r,mode))

ELSE (r/=round_under(r,mode))

ENDIF

If the under
ow trap is enabled, it is taken whenever tininess is detected. On a trapped under
ow,
the result must be scaled by �. Otherwise the both tininess and loss of accuracy must occur for
under
ow to be signaled. These cases are captured in the PVS de�nition of underflow:

underflow((r: nzreal | abs(r) < b ^ E_min), (mode: rounding_mode)):

[real, exception] =

IF tiny?(r, mode)

THEN IF trap_enabled?(underflow(FALSE))

THEN (round_scaled(r * b ^ alpha, mode),

underflow(exact_underflow(r, mode)))

ELSIF inaccurate?(r, mode) THEN (round_under(r, mode), underflow(TRUE))

ELSE (round_under(r, mode),

IF r = round_under(r, mode) THEN no_exceptions

ELSE inexact

ENDIF)

ENDIF

ELSE (round_under(r, mode), inexact)

ENDIF

This completes the de�nition of rounding in the presence of exceptions.

5.5 Inexact

The delivered result of a function does not change when the inexact exception is signalled. The
signal is raised whenever the value of the delivered result is di�erent from the in�nitely precise
intermediate result (i.e. inexact is signalled when rounding occurs). This can be computed with
respect to function round by using function is exact?.

Inexact may also be signaled in conjunction with over
ow or under
ow. These cases were
addressed above.

6 Traps

The PVS speci�cation does not address traps other than the declaration of the predicate trap enabled?

which is used to test whether a particular trap is enabled.

7 Concluding Remarks

This document described a partial de�nition of the IEEE-854 Standard for Radix-Independent
Floating-Point Arithmetic in the PVS veri�cation system. In most instances, there was a straight-
forward de�nition of the IEEE-854 features using the PVS speci�cation language. Formal tech-
niques are su�ciently mature that it is reasonable to consider use of formal speci�cation techniques
in the development of future standards.

The constraints enumerated in IEEE-854 for 
oating-point arithmetic are a generalization the
IEEE-754 Standard for Binary Floating-Point Arithmetic. Therefore, this formalization of the
IEEE-854 standard can be instantiated to serve as a basis for the formal speci�cation of IEEE-754
arithmetic. All that is required is to instantiate the general theory with the appropriate constants,
and de�ne the representation formats in accordance with IEEE-754.

The PVS theories described in this document provide a core formal basis for verifying any
proposed instance of IEEE 
oating-point arithmetic. We plan to explore the veri�cation of 
oating-
point systems with respect to the formal description presented here.

